

FUNCTIONAL PEARL

Inverting the Burrows–Wheeler transform

RICHARD S. BIRD and SHIN-CHENG MU

Programming Research Group, Oxford University, Wolfson Building, Parks Road, Oxford OX1 3QD, UK
(e-mail: richard.bird@comlab.ox.ac.uk)

Abstract

The Burrows–Wheeler Transform is a string-to-string transform which, when used as a preprocessing phase in compression, significantly enhances the compression rate. However, it often puzzles people how the inverse transform is carried out. In this pearl we to exploit simple equational reasoning to derive the inverse of the Burrows–Wheeler transform from its specification. We also outline how to derive the inverse of two more general versions of the transform, one proposed by Schindler and the other by Chapin and Tate.

1 Introduction

The Burrows–Wheeler Transform (Burrows & Wheeler, 1994) is a method for permuting a string with the aim of bringing repeated characters together. As a consequence, the permuted string can be compressed effectively using simple techniques such as move-to-front or run-length encoding. In Nelson (1996), the article that brought the BWT to the world’s attention, it was shown that the resulting compression algorithm could outperform many commercial programs available at the time. The BWT has now been integrated into the high-performance utility `bzip2`, available from Seward (2000).

Clearly, the best way of bringing repeated characters together is just to sort the string. But this idea has a fatal flaw as a preliminary to compression: there is no way to recover the original string unless the complete sorting permutation is produced as part of the output. Instead, the BWT achieves a more modest permutation, one that aims to bring some but not all repeated characters into adjacent positions. Moreover, the transform can be inverted using a single additional piece of information, namely an integer b in the range $0 \leq b < n$, where n is the length of the output (or input) string.

It often puzzles people, at least on a first encounter, as to why the BWT is invertible and how the inversion is actually carried out. We identify the fundamental reason why inversion is possible and use it to derive the inverse transform from its specification. As a bonus, we can further derive the inverse of two variations of the BWT transform, one proposed in Schindler (1997), another in Chapin & Tate (1998).

2 Defining the BWT

The BWT is specified by two functions: $bwp :: String \rightarrow String$, which permutes the string and $bwn :: String \rightarrow Int$, which computes the supplementary integer. The restriction to strings is not essential to the transform, and we can take bwp to have the Haskell type $Ord\ a \Rightarrow [a] \rightarrow [a]$, so lists of any type will do provided there is a total ordering relation on the elements. The function bwp is defined by

$$bwp = \text{map last} \cdot \text{lexsort} \cdot \text{rots} \quad (1)$$

The function $\text{lexsort} :: Ord\ a \Rightarrow [[a]] \rightarrow [[a]]$ sorts a list of lists into lexicographic order and is considered in greater detail in the following section. The function rots returns the rotations of a list and is defined by

$$\begin{aligned} \text{rots} &:: [a] \rightarrow [[a]] \\ \text{rots}\ xs &= \text{take} (\text{length}\ xs) (\text{iterate}\ \text{lrot}\ xs) \\ \text{lrot} &:: [a] \rightarrow [a] \\ \text{lrot}\ xs &= \text{tail}\ xs \# [\text{head}\ xs] \end{aligned}$$

The function lrot performs a single left rotation. In words, (1) reads: take the last column in the lexicographically sorted matrix of rotations of the input. The definition of bwp is constructive, but we won't go into details – at least, not in this pearl – as to how the program can be made more efficient.

The function bwn is specified by

$$\text{lexsort} (\text{rots}\ xs) !! \text{bwn}\ xs = xs \quad (2)$$

where $ys !! k$ returns the element of ys in position k . In words, $\text{bwn}\ xs$ returns some position at which xs occurs in the sorted list of rotations of xs . If xs is a repeated string, then $\text{rots}\ xs$ will contain duplicates, so $\text{bwn}\ xs$ is not defined uniquely by (2).

As an illustration, consider the string `yokohama`. The rotations and the lexicographically sorted rotations are as follows:

y o k o h a m a	0 a m a y o k o h
o k o h a m a y	1 a y o k o h a m
k o h a m a y o	2 h a m a y o k o
o h a m a y o k	3 k o h a m a y o
h a m a y o k o	4 m a y o k o h a
a m a y o k o h	5 o h a m a y o k
m a y o k o h a	6 o k o h a m a y
a y o k o h a m	7 y o k o h a m a

The output of bwp is the string `hmooakya`, the last column of the second matrix, and $\text{bwn}\ \text{"yokohama"} = 7$ because row number 7 in the sorted matrix of rotations is the input string.

The BWT helps compression because it brings together characters with a common context. To give a brief illustration, an English text may contain many occurrences of words such as “this”, “the”, “that” and some occurrences of “where”, “when”, “she”, “he” (with a space), etc. Consequently, many of the rotations beginning with

“h” will end with a “t”, some with a “w”, an “s” or a space. The chance is smaller that a rotation beginning with “h” would end in a “x”, a “q”, or an “u”, etc. Thus the BWT brings together a subset of the letters, say, those “t”s, “w”s and “s”s. A move-to-front encoding phase is then able to convert the characters into a series of small-numbered indexes, which improves the effectiveness of entropy-based compression techniques such as Huffman or arithmetic coding. For a fuller understanding of the role of the BWT in data compression, consult Burrows & Wheeler (1994) and Nelson (1996).

The inverse transform $unbwt :: Ord\ a \Rightarrow Int \rightarrow [a] \rightarrow [a]$ is specified by

$$unbwt\ (bwn\ xs)\ (bwp\ xs) = xs \tag{3}$$

To compute $unbwt$ we have to show how the matrix of lexicographically sorted rotations, or at least row number t , where $t = bwn\ xs$, can be recreated solely from the knowledge of its last column. To do so we need to examine lexicographic sorting in more detail.

3 Lexicographic sorting

Let $(\leq) :: a \rightarrow a \rightarrow Bool$ be a linear ordering on a . Define $(\leq_k) :: [a] \rightarrow [a] \rightarrow Bool$ inductively by

$$\begin{aligned} xs \leq_0 ys &= True \\ (x : xs) \leq_{k+1} (y : ys) &= x < y \vee (x = y \wedge xs \leq_k ys) \end{aligned}$$

The value $xs \leq_k ys$ is defined whenever the lengths of xs and ys are both no smaller than k .

Now, let $sort\ (\leq_k) :: [[a]] \rightarrow [[a]]$ be a stable sorting algorithm that sorts an $n \times n$ matrix, given as a list of lists, according to the ordering \leq_k . Thus $sort\ (\leq_k)$, which we henceforth abbreviate to $sort\ k$, sorts a matrix on its first k columns. Stability means that rows with the same first k elements appear in their original order in the output matrix. By definition, $lexsort = sort\ n$.

Define $cols\ j = map\ (take\ j)$, so $cols\ j$ returns the first j columns of a matrix. Our aim in this section is to establish the following fundamental relationship, which is the key property for establishing the existence of an algorithm for inverting the BWT. Provided $1 \leq j \leq k$, we have

$$cols\ j \cdot sort\ k \cdot rrots = sort\ 1 \cdot cols\ j \cdot map\ rrot \cdot sort\ k \cdot rrots \tag{4}$$

The function $rrot$ denotes a single right rotation, defined by:

$$rrot\ xs = last\ xs : init\ xs$$

Equation (4) looks daunting, but take $j = n$ (so $cols\ j$ is the identity) and $k = n$ (so $sort\ k$ is a complete lexicographic sorting). Then (4) states that the following transformation on the sorted rotations is the identity: move the last column to the front and re-sort the rows on the new first column. As we will see, this implies that the permutation that produces the first column from the last column is the same as that which produces the second from the first, and so on.

To prove (4) we will need some basic properties of rotations and sorting. For rotations, one identity suffices:

$$\text{map rrot} \cdot \text{rots} = \text{rrot} \cdot \text{rots} \quad (5)$$

More generally, applying a rotation to the columns of a matrix of rotations has the same effect as applying the same rotation to the rows.

For sorting we will need

$$\text{sort } k \cdot \text{map rrot}^k = (\text{sort } 1 \cdot \text{map rrot})^k \quad (6)$$

where f^k is the composition of f with itself k times. Equivalently, equation (6) can be read as $\text{sort } k = (\text{sort } 1 \cdot \text{map rrot})^k \cdot \text{map lrot}^k$. This identity formalises the fact that one can sort a matrix on its first k columns by first rotating the matrix to bring these columns into the last k positions, and then repeating k times the process of rotating the last column into first position and *stably* sorting according to the first column only. Since $\text{map rrot}^n = \text{id}$, the initial processing is omitted in the case $k = n$, and we have the standard definition of *radix sort*. In this context see Gibbons (1999) which deals with the derivation of radix sorting in a more general setting.

Substituting $k + 1$ for k in (6) and expanding the right-hand side, we obtain

$$\text{sort } (k + 1) \cdot \text{map rrot}^{k+1} = \text{sort } 1 \cdot \text{map rrot} \cdot \text{sort } k \cdot \text{map rrot}^k$$

Since $\text{rrot}^k \cdot \text{rrot}^{n-k} = \text{rrot}^n = \text{id}$ we can compose both sides with map rrot^{n-k} to obtain

$$\text{sort } (k + 1) \cdot \text{map rrot} = \text{sort } 1 \cdot \text{map rrot} \cdot \text{sort } k \quad (7)$$

Finally, we shall need the following two properties of columns. First, for arbitrary j and k :

$$\text{cols } j \cdot \text{sort } k = \text{cols } j \cdot \text{sort } (j \text{ \textbf{min} } k) = \text{sort } (j \text{ \textbf{min} } k) \cdot \text{cols } j \quad (8)$$

In particular, $\text{cols } j \cdot \text{sort } k = \text{cols } j \cdot \text{sort } j$ whenever $j \leq k$. Furthermore, since $\text{sort } k$ sorts the list of strings by the first k characters only, we have:

$$\text{cols } j \cdot \text{sort } k \cdot \text{perm} = \text{cols } j \cdot \text{sort } k \quad (9)$$

whenever $j \leq k$ and perm is any function that permutes its argument.

Having introduced the fundamental properties (5), (7), (8) and (9), we can now prove (4). With $1 \leq j \leq k$ we reason:

$$\begin{aligned} & \text{sort } 1 \cdot \text{cols } j \cdot \text{map rrot} \cdot \text{sort } k \cdot \text{rots} \\ &= \quad \{\text{by (8)}\} \\ & \text{cols } j \cdot \text{sort } 1 \cdot \text{map rrot} \cdot \text{sort } k \cdot \text{rots} \\ &= \quad \{\text{by (7)}\} \\ & \text{cols } j \cdot \text{sort } (k + 1) \cdot \text{map rrot} \cdot \text{rots} \\ &= \quad \{\text{by (8)}\} \\ & \text{cols } j \cdot \text{sort } k \cdot \text{map rrot} \cdot \text{rots} \\ &= \quad \{\text{by (5)}\} \end{aligned}$$

$$\begin{aligned}
 & \text{cols } j \cdot \text{sort } k \cdot \text{rrot} \cdot \text{rots} \\
 = & \quad \{\text{by (9)}\} \\
 & \text{cols } j \cdot \text{sort } k \cdot \text{rots}
 \end{aligned}$$

Thus, (4) is established.

4 The derivation

Our aim is to develop a program that reconstructs the sorted matrix from its last column. In other words, we aim to construct $\text{sort } n \cdot \text{rots} \cdot \text{unbwt } t$. In fact, we will try to construct a more general expression $\text{cols } j \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t$ (of which the former expression is the case $j = k = n$) because the more general expression is used in the two variants of the BWT described in sections 5 and 6.

First, observe that for $0 \leq j$,

$$\text{cols } (j + 1) \cdot \text{map } \text{rrot} = \text{join} \cdot \text{fork } (\text{map } \text{last}, \text{cols } j) \quad (10)$$

where $\text{join } (xs, xss) = \text{zipWith } (:) xs xss$ is the matrix xss with xs adjoined as a new first column, and $\text{fork } (f, g) x = (f x, g x)$. Hence

$$\begin{aligned}
 & \text{cols } (j + 1) \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t \\
 = & \quad \{\text{by (4)}\} \\
 & \text{sort } 1 \cdot \text{cols } (j + 1) \cdot \text{map } \text{rrot} \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t \\
 = & \quad \{\text{by (10)}\} \\
 & \text{sort } 1 \cdot \text{join} \cdot \text{fork } (\text{map } \text{last}, \text{cols } j) \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t \\
 = & \quad \{\text{since } \text{fork } (f, g) \cdot h = \text{fork } (f \cdot h, g \cdot h)\} \\
 & \text{sort } 1 \cdot \text{join} \cdot \text{fork } (\text{map } \text{last} \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t, \\
 & \quad \text{cols } j \cdot \text{sort } k \cdot \text{rots} \cdot \text{unbwt } t)
 \end{aligned}$$

In particular, consider $t = \text{bwn } xs$ for an input xs and $k = n$, the length of xs . Since $\text{bwp} = \text{map } \text{last} \cdot \text{sort } n \cdot \text{rots}$, and $\text{bwp } (\text{unbwt } t \text{ } xs) = xs$, the equality shown above reduces to:

$$\begin{aligned}
 & (\text{cols } (j + 1) \cdot \text{sort } n \cdot \text{rots} \cdot \text{unbwt } t) \text{ } xs \\
 & = (\text{sort } 1 \cdot \text{join} \cdot \text{fork } (\text{id}, \text{cols } j \cdot \text{sort } n \cdot \text{rots} \cdot \text{unbwt } t)) \text{ } xs
 \end{aligned}$$

Setting $\text{recreate } j = \text{cols } j \cdot \text{sort } n \cdot \text{rots} \cdot \text{unbwt } t$, we have just constructed a recursive definition for recreate :

$$\begin{aligned}
 \text{recreate } 0 & = \text{map } (\text{const } []) \\
 \text{recreate } (j + 1) & = \text{sort } 1 \cdot \text{join} \cdot \text{fork } (\text{id}, \text{recreate } j)
 \end{aligned}$$

The Haskell code for recreate is given in Figure 1. The function $\text{sortBy} :: (a \rightarrow a \rightarrow \text{Bool}) \rightarrow [a] \rightarrow [a]$ is a stable sorting algorithm. It is identical to the standard function sortBy except for a slightly different type.

Now we know that recreate reconstructs the entire matrix, we just need to pick a particular row. Taking $j = n$, we have $\text{unbwt } t = (!! t) \cdot \text{recreate } n$. The problem is that this implementation of unbwt involves computing $\text{sort } 1$ a total of n times. To

```

recreate :: Ord a => Int -> [a] -> [[a]]
recreate 0    = map (const [])
recreate (j+1) = sortby leq . join . fork (id, recreate j)
  where leq us vs = head us <= head vs
        join = uncurry (zipWith (:))
        fork (f,g) x = (f x, g x)

```

Fig. 1. Computation of *recreate*.

avoid repeated sorting, observe that $recreate\ 1\ ys = sort\ ys$, where the function *sort* now sorts a list rather than a matrix of one column. Furthermore, for some suitable permutation *sp* we have

$$sort\ ys = permby\ sp\ ys$$

where $permby :: (Int \to Int) \to [a] \to [a]$ applies a permutation to a list:

$$permby\ p\ [x_0, \dots, x_{n-1}] = [x_{p(0)}, \dots, x_{p(n-1)}]$$

It follows that

$$recreate\ (j + 1)\ ys = join\ (permby\ sp\ ys, permby\ sp\ (recreate\ j\ ys))$$

Equivalently,

$$recreate\ n\ ys = transpose\ (take\ n\ (iterate1\ (permby\ sp)\ ys)) \quad (11)$$

where $transpose :: [[a]] \to [[a]]$ is the standard Haskell function for transposing a matrix and $iterate1 = tail \cdot iterate$. The t th row of a matrix is the t th column of the transposed matrix, that is, $(!!t) \cdot transpose = map\ (!!t)$, so we can use the naturality of $take\ n$ to obtain

$$unbwt\ t\ ys = take\ n\ (map\ (!!t)\ (iterate1\ (permby\ sp)\ ys))$$

Suppose we define $spl :: Ord\ a \Rightarrow [a] \to Int \to (a, Int)$ by

$$spl\ ys\ i = sort\ (zip\ ys\ [1..])\ !!\ i$$

where *sort* now sorts a list of pairs. Then

$$spl\ ys\ j = (ys\ !!\ sp\ j, sp\ j)$$

Hence

$$map\ (!!k)\ (iterate1\ (permby\ sp)\ ys) = thread\ (spl\ ys\ k)$$

where $thread\ (x, j) = x : thread\ (spl\ ys\ j)$.

The final algorithm, written as a Haskell program, is given in Figure 2. In a real implementation, the sorting in *spl* would be performed by counting the histogram of the input, which can be done in linear time using a mutable array. The “threading” part can be performed in linear time, assuming constant-time array look-up. In Seward (2001) it was observed that the main inefficiency with the algorithm lies in the cache misses involved in the threading, arising as a result of accessing a large array in non-sequential order.

```

unbwt :: Ord a => Int -> [a] -> [a]
unbwt t ys = take (length ys) (thread t)
  where spl i = sort (zip ys [0..]) !! i
        thread i = x : thread j
          where (x,j) = spl i

```

Fig. 2. Computation of *unbwt*.

5 Schindler's variation

The main variation of BWT is to exploit the general form of (4) rather than the special case $k = n$. Suppose we define

$$bwpS\ k = \text{map last} \cdot \text{sort } k \cdot \text{rots}$$

This version, which sorts only on the first k columns of the rotations of a list, was considered in Schindler (1997). The derivation of the previous section shows how we can recreate the first k columns of the sorted rotations from $ys = bwp\ k\ xs$, namely by computing *recreate* $k\ ys$.

The remaining columns cannot be computed in the same way. However, we can reconstruct the t th row, where $t = bwn\ k\ xs$ and

$$\text{sort } k (\text{rots } xs) !! t = xs$$

The first k elements of xs are given by *create* $k\ ys\ !!\ t$, and the last element of xs is $ys\ !!\ t$. Certainly we know

$$\text{take } k (\text{rrot } xs) = [x_n, x_1, \dots, x_{k-1}]$$

This list begins with the *last* row of the unsorted matrix, and consequently, since sorting is stable, will be the *last* occurrence of the list in *create* $k\ ys$. If this occurrence is at position p , then $ys\ !!\ p = x_{n-1}$. Having discovered x_{n-1} , we know *take* $k\ (\text{rrot}^2\ xs)$. This list begins the penultimate row of the unsorted matrix, and will be either the last occurrence of the list in the sorted matrix, or the penultimate one if it is equal to the previous list. We can continue this process to discover all of $[x_{k+1}, \dots, x_n]$ in reverse order. Efficient implementation of this phase of the algorithm requires building an appropriate data structure for repeatedly looking up elements in reverse order in the list *zip* (*recreate* $k\ ys$) ys and removing them when found. A simple implementation is given in Figure 3.

6 Chapin and Tate's variation

Primarily for the purpose of showing that the pattern of derivation in this paper can be adapted to other cases, we shall consider another variation. Define the following alternative of BWT:

$$bwpCT\ k = \text{map last} \cdot \text{twist}^k \cdot \text{lexsort} \cdot \text{rots}$$

where the function *twist* rearranges the rows of the matrix. One possible choice

```

unbwt :: Ord a => Int -> Int -> [a] -> [a]
unbwt k p ys = us ++ reverse (take (length ys - k) vs)
  where us = yss !! p
        yss = recreate k ys
        vs = u:search k (reverse (zip yss ys)) (take k (u:us))
        u = ys !! p

search :: Eq a => Int -> [[a],a] -> [a] -> [a]
search k table xs = x:search k table' (take k (x:xs))
  where (x,table') = dlookup table xs

dlookup :: Eq a => [(a,b)] -> a -> (b,[(a,b)])
dlookup ((a,b):abs) x = if a==x then (b,abs)
  else (c,(a,b):cbs)
  where (c,cbs) = dlookup abs x

```

Fig. 3. Computation of Schindler's variation.

```

twist :: Eq a => Int -> [[a]] -> [[a]]
twist k = concat . mapEven (map reverse) . groupby (take k)

mapEven, mapOdd :: (a->a) -> [a] -> [a]
mapEven f [] = []
mapEven f (x:xs) = f x : mapOdd f xs
mapOdd f [] = []
mapOdd f (x:xs) = x : mapEven f xs

```

Fig. 4. One possible choice of *twist*.

of *twist* is shown in Figure 4. As an example, consider the rotations of the string aabab:

aabab	ababa	abaab
abaab	abaab	ababa
ababa	aabab	aabab
baaba	baaba	babaa
babaa	babaa	baaba

Shown on the left is the sorted matrix of rotations. The matrix in the middle is the result of applying *twist*. The rows are first partitioned into groups by *groupby* according to their first characters. The even numbered groups (counting from zero) are then reversed. In the example, the group starting with a is reversed. Shown on the right is the result of applying *twist*² to the matrix in the middle. The rows are partitioned into three groups, starting with ab, aa, and ba respectively. The noughth and the second group are reversed.

The idea of twisting the matrix of sorted rotations was proposed in Chapin & Tate (1998), where a similar but slightly more complicated version of *twist* was considered based on the Gray code. Chapin and Tate's generalisation can marginally improve the compression ratio of the transformed text.

For twist^k to be invertible, however, we need only the property that for $0 < j \leq k$,

$$\text{cols } j \cdot \text{twist}^k = \text{cols } j \cdot \text{twist}^{j-1} \quad (12)$$

In words, further twisting (twist^k where $j \leq k$) does not change the first j columns after they have been set by twist^{j-1} . In the example above, for instance, the call to twist^2 does not change the first two columns of the matrix in the middle, nor do successive calls to twist^k where $k \geq 2$. Any twist satisfying (12) suffices to make bwtCT invertible. This separation of concerns on compression rate and invertibility means that one can try many possible choices satisfying (12) and experiment with the effect on compression.

To derive an algorithm for the reverse transform we need the following analogue of (4):

$$\begin{aligned} & \text{cols } j \cdot \text{twist}^k \cdot \text{lexsort} \cdot \text{rots} \\ = & \text{twist}^k \cdot \text{sort } 1 \cdot \text{untwist}^k \cdot \text{cols } j \cdot \text{map rrot} \cdot \text{twist}^k \cdot \text{lexsort} \cdot \text{rots} \end{aligned} \quad (13)$$

where untwist is inverse to twist . The proof of (13) follows a similar path to the derivation in section 3. When $k = 0$ (so $\text{twist}^k = \text{id}$), equation (13) reduces to a special case of (4). In words, (13) means that the following operation is an identity on a matrix generated by $\text{twist}^k \cdot \text{lexsort} \cdot \text{rots}$: move the last column to the first, untwist it, sort it by the first character, and twist it again.

Based on (13) one can now derive an algorithm similar to that of section 4. Defining

$$\text{recreateCT } j \ k = \text{cols } j \cdot \text{twist}^k \cdot \text{lexsort} \cdot \text{rots} \cdot \text{unbwtCT } t$$

we can construct a recursive definition for recreateCT which is similar to (11), but with the permutation sp simulating $\text{twist}^i \cdot \text{sort } 1 \cdot \text{untwist}^i$ for appropriate i , rather than just $\text{sort } 1$. The details are more complicated than for the corresponding definition of recreate (which builds one column in each step) because in recreateCT the permutation sp changes each time a new column is built. So the algorithm has to construct a new permutation as well as a new column at each step. The resulting algorithm will thus return a pair whose first component is the reconstructed matrix and the second component is a permutation representing sp . In the first step we build the first column and a permutation simulating $\text{twist} \cdot \text{sort } 1 \cdot \text{untwist}$; in the second step we build the second column and a permutation for $\text{twist}^2 \cdot \text{sort } 1 \cdot \text{untwist}^2$, and so on. Further details are omitted in this pearl.

7 Conclusions

We have shown how the inverse Burrows–Wheeler transform can be derived by equational reasoning. The derivation can be re-used to invert the more general versions proposed by Schindler, and by Chapin and Tate.

Other aspects of the BWT also make interesting topics. In Manber & Myers (1993) it is shown how to sort the rotations of a given string in $O(n \log n)$ time using suffix arrays, where n is the length of the string. How efficiently it can be done in

a functional setting remains unanswered, though we conjecture that $O(n(\log n)^2)$ is the best possible.

Acknowledgements

We would like to thank Ian Bayley, Jeremy Gibbons, Ralf Hinze, Geraint Jones and Barney Stratford for constructive criticism of earlier drafts of this pearl, and to Julian Seward who made some useful comments regarding the practical aspects of the Burrows-Wheeler transform. Also, we would like to thank two anonymous referees for their detailed and useful comments, and for identifying many typos and infelicities in an earlier draft of this paper.

References

- Burrows, M. and Wheeler, D. J. (1994) *A block-sorting lossless data compression algorithm*. Technical report, Digital Systems Research Center. Research Report 124. (Available online at <http://gatekeeper.dec.com/pub/DEC/SRC/research-reports/abstracts/src-rr-124.html>.)
- Chapin, B. K. and Tate, S. (1998) Higher compression from the Burrows-Wheeler transform by modified sorting. *Data Compression Conference 1998*, p. 532. IEEE Press.
- Gibbons, J. (1999) A pointless derivation of radixsort. *J. Functional Program.* **9**(3), 339–346.
- Manber, U. and Myers, G. (1993) Suffix arrays: A new method for on-line string searches. *SIAM J. Comput.* **22**(5), 935–948.
- Nelson, M. (1996) Data compression with the Burrows-Wheeler transform. *Dr. Dobb's J.* September.
- Schindler, M. (1997) A fast block-sorting algorithm for lossless data compression. *Data Compression Conference 1997*, p. 469. IEEE Press.
- Seward, J. (2000) bzip2. <http://sources.redhat.com/bzip2/>.
- Seward, J. (2001) Space-time tradeoffs in the inverse B-W transform. *Data Compression Conference 2001*, pp. 439–448.